

FlexPDE v7 Trial Script [E]

本トライアルでは 3 次元における熱伝導の問題について考察してみます。FlexPDE において 3 次元オブジェクトを構成する場合、基本となるのは $x-y$ 平面上のリージョン定義を z 軸方向に押し出す (extrude) という考え方です。従って垂直な角柱や円柱からなるオブジェクトの設定は比較的容易に行えるわけですが、球面とか円錐面を含むオブジェクトの場合、extrusion の考え方に対応できるのだろうかという疑問を持たれるかも知れません。ここでは円錐面を含むケースについてスクリプトを作成してみます。

1 問題設定

図 1 は円筒状の容器の断面図を示したものです。




図 1 円筒状容器の断面図

容器の上には円錐台状の凹部を設定してあります。図で赤く示した面を 100°C 、容器の底面を 0°C に維持したときの温度分布を FlexPDE によって解析してみます。なお、図 1 上でシェーディングを施した面は断熱されているものとします。オブジェクトの形状にかかわるパラメータとしては図 1 に示した r_0, r_a, r_b, h_0, h_1 があるわけですが、スクリプト上では次のような寸法を想定しています。

$$\begin{aligned} r_0 &= 8\text{cm} & r_a &= 5\text{cm} & r_b &= 3\text{cm} \\ h_0 &= 8\text{cm} & h_1 &= 4\text{cm} \end{aligned}$$

2 上面の定義

このような3次元構成の場合、上面の規定が問題となるわけですが、基本となるのは x - y 平面上のリージョンごとにその上面を定義して行くという考え方です。図2は x - y 平面上のリージョン構成を示したものです。




図2 リージョン構成

同心円状に次のような3つのリージョンを想定することになります。なお、これらのリージョンは重なった形で設定されることになるわけですが、その場合、重複した部分については後から定義した情報が有効となります。そのため定義順序が重要な意味を持つことになる点に注意してください。

- Region1 – 解析対象ドメインを表す半径 r_0 の円形領域。このリージョンにおける上面の方程式は $z = h_0$ で与えられる。
- Region2 – 円錐台上面に対応する半径 r_a の円形領域。このリージョンにおける上面、すなわち円錐面の方程式は図1に示した z, r を用いて

$$z = h_1 + (r - r_b) \cdot \tan \quad (1)$$

のように記述される。ただし

$$r = \sqrt{x^2 + y^2} \quad \tan = \frac{h_0 - h_1}{r_a - r_b} \quad (2)$$

である。

- Region3 – 円錐台下面に対応する半径 r_b の円形領域。このリージョンにおける上面の方程式は $z = h_1$ で与えられる。

3 スクリプト

ここでは次のようなスクリプト trial7e.pde を用意します。

```

TITLE
  '3D Heat Flow'

COORDINATES
  cartesian3

VARIABLES
  U      { temperature }

DEFINITIONS
  k = 1      { conductivity }
  r0 = 0.08   ra = 0.05   rb = 0.03
  h0 = 0.08   h1 = 0.04
  tantheta = (h0 - h1)/(ra - rb)
  surf1      { upper surface to be defined regionally }

EQUATIONS
  div(k*grad(U)) = 0

EXTRUSION
  SURFACE 'bottom'   z = 0
  SURFACE 'top'     z = surf1

BOUNDARIES
  SURFACE 'bottom' VALUE(U) = 0

  REGION 1      { total domain }
    surf1 = h0
    SURFACE 'top' NATURAL(U) = 0
    START (r0, 0) NATURAL(U) = 0
      ARC(CENTER=0,0) ANGLE=360 TO CLOSE

  REGION 2      { circle with radius ra }
    surf1 = h1 + (sqrt(x^2 + y^2) - rb)*tantheta
    SURFACE 'top' VALUE(U) = 100
    START (ra, 0)
      ARC(CENTER=0,0) ANGLE=360 TO CLOSE

```

```

REGION 3      { circle with radius rb }
surf1 = h1
SURFACE 'top' VALUE(U) = 100
START (rb, 0)
      ARC(CENTER=0,0) ANGLE=360 TO CLOSE

```

PLOTS

```

GRID(x,y,z) paintregions
GRID(x,z) on y = 0 paintregions
CONTOUR(U) on y = 0 painted
SURFACE(U) on z = 0.9*h1 fixed range (0, 100)

```

END

セクションごとに説明を補足しておくようになります。

(1) COORDINATES セクション

座標系としては 3 次元直交座標系を意味する cartesian3 という指定を行います。本セクションを省略すると 2 次元直交座標系 cartesian2 が仮定されることとなります。

(2) DEFINITIONS セクション

関連するパラメータの値や式を定義します。なお、上面の数式に対応する surf1 というパラメータについては定義がリージョンごとに行われることになるため、ここでは名称の登録のみを行っています。

(3) EQUATIONS セクション

EQUATIONS セクションでは熱伝導方程式を記述します。なお、熱伝導率を意味する k の値は 3D オブジェクト内で一様であり、DEFINITIONS セクション内で $k = 1$ という値を設定しています。

(4) EXTRUSION セクション

3 次元の場合に必要なセクションで、 z 軸方向の面やレイヤの構成定義を下から順に行います。ここでは容器下面と上面をそれぞれ 'bottom', 'top' としてその数式を指定しています。ただし上面についてはリージョンごとに数式が異なるため、単に $z = \text{surf1}$ という指定となっています。

(5) BOUNDARIES セクション

BOUNDARIES セクションでは x - y 平面上におけるリージョン構成の規定が中心となりますが、同時に境界条件の設定も行います。最初の Surface 文では容器下面における境界条件 $U = 0$ を指定しています。


続いてリージョンごとにその形状と surf1 の数式を規定します。なお、最も外側に位置する Region 1 の場合には上面が絶縁境界となるため、Surface 文で $\frac{\partial U}{\partial n} = 0$ という指定を行っていますが、それ以外に円周境界上でも $\frac{\partial U}{\partial n} = 0$ という指定を加えている点に注意してください。この指定があるため、Region 1 を extrude し構成される円柱側面も絶縁境界となります。

Region 2 の定義中では surf1 に対する数式として (1) 式が与えられています。また上面の境界条件を $U = 100$ と設定しています。同様に Region 3 の定義中では surf1 = h1、上面の境界条件として $U = 100$ という指定を行っています。

4 実行結果


(1) GRID(x,y,z) paintregions

設定された 3D オブジェクトの表面上におけるメッシュ構成を示すプロットです。リージョン別に色分けした形の表示となるよう、オプション paintregions を指定しています。




(2) GRID(x,z) on $y = 0$ paintregions

平面 $y = 0$ という切断面上におけるメッシュ構成をリージョン別に色分けしてプロットしたものです。図形の設定が正しく行われているかどうかをチェックするのに有効です。

(3) CONTOUR(U) on $y = 0$ painted

平面 $y = 0$ 上における温度分布の様子を色塗り等高線図の形でプロットしたものです。等温線は絶縁壁に対して直交していることが見て取れます。



(4) SURFACE(U) on $z = 0.9h_1$ fixed range (0, 100)

円錐台底面よりわずかに離れた水平面 $z = 0.9h_1$ を切断面としたときの温度分布を 3 次元曲面図の形で表示させたものです。なお、カラースキームが (3) の場合と共通のものとなるよう、fixed range (0,100) というオプションを指定している点にご注意ください。

